/*
Copyright (c) 2009, Young Digital Planet SA
All rights reserved.

* Redistribution and use in source and binary forms, with or without 
  modification, are permitted provided that the following conditions are met:
* Redistributions of source code must retain the above copyright notice, this 
  list of conditions and the following disclaimer.
* Redistributions in binary form must reproduce the above copyright notice, 
  this list of conditions and the following disclaimer in the documentation 
  and/or other materials provided with the distribution.
* Neither the name of the Young Digital Planet SA nor the names of its 
  contributors may be used to endorse or promote products derived from this 
  software without specific prior written permission.

THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" 
AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE 
IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE 
ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE 
LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR 
CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF 
SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS 
INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN 
CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE) 
ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE 
POSSIBILITY OF SUCH DAMAGE.
*/
package pl.ydp.utils
{
	/**
	 * Narzędzia operujące na ścieżkach do plików
	 */
	public class UriUtil
	{
		/** 
		 * Likwiduje wszelkie odwołania względne w ścieżce (/../)
		 * @param path_str ścieżka
		 * @return ścieżka, ze zredukowanymi odwołaniami względnymi, jeżeli 
		 * odwołanie względne .. jest pierwszym elementem ścieżki to nie 
		 * zostanie usunięte.
		 */
		public static function resolveRelatives( path_str:String ):String {
			path_str = UriUtil.normalizeURL( path_str );
			path_str = path_str.split( "\\" ).join( "/" );
			path_str = path_str.split( "/./" ).join( "/" );

			var protoAndServer:String = "";
			// teraz oddzielam proto i serwer [http://costam/]......
			var cix:int = path_str.indexOf(":");
			if ( cix>0 ) {
				++cix;	// (1) cix ma wskazywać na pierwszy znak nie należący do proto/server
				while( path_str.charAt(cix)=='/' )
					++cix;
				var cix2:int = path_str.indexOf('/', cix );
				if ( cix2!=-1 )
					cix = cix2+1;	// (1)
				protoAndServer = path_str.substring( 0, cix );
				path_str = path_str.substring( cix );
			}
			// oddzielam query string
			var query:String = "";
			cix = path_str.indexOf("?");
			if ( cix>=0 ) {
				query = path_str.substring( cix );
				path_str = path_str.substring( 0, cix );
			}
			// przenoszę elementy ścieżki do final_array. Gdy trafiam
			// na .. to zdejmuję ostatnio wstawiony element w final_array
			var path_array:Array = path_str.split( "/" );
			var final_array:Array = [];
			for( var ix:int = 0; ix<path_array.length; ++ix ) {
				if ( path_array[ix]==".." && final_array.length>0 && final_array[final_array.length-1]!=".." )
					final_array.pop();
				else
					final_array.push( path_array[ix] );
			}
			
			return protoAndServer+final_array.join("/")+query;
		}
		
		/**
		 * Sprawdza czu podana ścieżka jest absolutna (zawiera protokół)
		 * @param uri ściezka do sprawdzenia
		 * @return true jeżeli ścieżka jest absolutna
		 */
		public static function isAbsolutePath(uri:String):Boolean{
			return ( uri.toLowerCase().indexOf("http:")==0 ||
				 		uri.toLowerCase().indexOf("file:")==0 ||
				 		uri.toLowerCase().indexOf("app:")==0) 
		}
		
		/**
		 * Wycina ze ścieżki nazwę pliku bez rozszerzeń
		 * @param src ścieżka do pliku
		 * @return nazwę pliku bez rozszerzeń
		 */
		public static function filename( src:String ):String {
			src = UriUtil.normalizeURL( src );
			var ind1:int = src.lastIndexOf("/")+1;
			var ind2:int = src.lastIndexOf("\\")+1;
			var ind3:int = Math.max(ind1, ind2);
			var res:String = src.substr( Math.max(0,ind3), src.length );
			var ind4:int = res.lastIndexOf(".");
			if ( -1 != ind4) {
				return res.substring(0, ind4);
			}
			return res;
		}
		
		/**
		 * Wycina ze ścieżki nazwę pliku z rozszerzeniem
		 * @param	src
		 * @return nazwa pliku z rozszerzeniem
		 */
		public static function fullFilename( src:String ):String {
			src = UriUtil.normalizeURL( src );
			var ind1:int = src.lastIndexOf("/")+1;
			var ind2:int = src.lastIndexOf("\\")+1;
			var ind3:int = Math.max(ind1, ind2);
			var path:String = src.substr( Math.max(0, ind3), src.length );
			ind1 = path.indexOf("?");
			return (ind1!=-1)? path.substring(0, ind1) : path;
		}
		
		/**
		 * Wycina ze ścieżki rozszerzenie pliku
		 * @param src ścieżka do pliku
		 * @return rozszerzenie pliku
		 */
		public static function extension( path:String ):String {
			path = UriUtil.normalizeURL( path );
			var ind1:int = path.lastIndexOf(".");
			if (-1==ind1) {
				return "";
			}
			path = path.substring(ind1 + 1, path.length);
			var ind2:int = path.indexOf("?");
			return (ind2!=-1)? path.substring(0, ind2) : path;
		}
		
		/**
		 * Wycina ze ścieżki katalog
		 * @param src ściezka do pliku
		 * @return katalog wraz z nazwą serwera
		 */
		public static function dirname(path:String):String {
			path = UriUtil.normalizeURL( path );
			var ind1:int, ind2:int, ind3:int, ind4:int, ind5:int, ind6:int;
			ind1 = path.lastIndexOf("\\");
			ind2 = path.lastIndexOf("/");
			ind3 = path.lastIndexOf(":");
			var ind_1:int = Math.max(ind1, Math.max(ind2, ind3));
			ind4 = path.lastIndexOf(escape("\\"));
			ind5 = path.lastIndexOf(escape("/"));
			ind6 = path.lastIndexOf(escape(":"));
			var ind_3:int = Math.max(ind4, Math.max(ind5, ind6));
			var last:int = (ind_1 < ind_3)? ind_3 + 3 : ind_1 + 1;
			return path.substring(0, last);
		}
		
		/**
		 * Usuwa ze ściezki informację o RSL'sach [DYNAMIC].
		 * @see mx.utils.LoaderUtil#normalizeURL()
		 */
		public static function normalizeURL(path:String):String {
			// usuwam ze ścieżki znacznik dodawany do urla aplikacji używającej RSL 
			var result:String = path.split("/[[DYNAMIC]]/")[0];
			return result;
		}
	}
}